



(<https://wrds->

[web.wharton.upenn.edu/wrds/](https://web.wharton.upenn.edu/wrds/))

[🏠 \(/wrds/index.cfm\)](#) / [Research \(/wrds/research/index.cfm\)](#)

/ [Applications \(/wrds/research/applications/index.cfm\)](#)

/ [Microstructure \(/wrds/research/applications/microstructure/index.cfm\)](#)

/ [NBBO Derivation \(/wrds/research/applications/microstructure/NBBO\\_derivation/index.cfm\)](#)

/ [Research Tools](#)

## NBBO Derivation using SAS Data Views

This sample program allows users to efficiently calculate the best bid and offer prices and depths at those prices (the “inside quote”) from TAQ Quote data, whenever research requires the use of inside quotes at a particular point in time. This macro illustrates the efficiency of using SAS DATA VIEWS to compute the National Best Bid and Offer (NBBO) for NYSE’s Trade and Quote (TAQ) data. We encourage users to build on the concepts in this program, mainly the employment of SAS data views, as they will save you some programming effort and your programs will run faster by using fewer system resources. If you want to run the code as is, WRDS has stored all the output datasets that contain the NBBOs at each second for all stocks with quotes data in TAQ since 1993, in the following directory on WRDS-CLOUD server: [/wrds/nyse/sasdata/wrds\\_taq\\_nbbo/](#).

### SAS Views:

A SAS data VIEW is conceptually like a SAS data SET. In the sample NBBO program, both are created in a data statement and both could also be created in a PROC SQL. However, unlike a SAS data SET, a data VIEW merely establishes programming logic when the data statement is encountered. It does not immediately process the data as would a data SET declaration. The data VIEW actually processes the data only when the view is accessed in a subsequent PROC or DATA step. That is, the view, instead of writing data to disk, only to be retrieved from disk in a later step, essentially “pipes” the data directly to the subsequent step, thereby avoiding superfluous disk input/output (I/O) activity and shortening the program’s time to completion. The “price” of the reduced disk I/O is additional system memory use when accessing the view. Typically the time savings makes the additional memory requirement worthwhile, although a single deep “nesting” of views (i.e. views accessing other views) can be less effective than a couple of shallow nestings.

### Quote Rule:

Exchange specialists or market makers have special functions in the securities markets because they trade for their own accounts while also handling orders for customers, and are required to comply with specific SEC rules, regarding the publishing of quotes and handling customer orders. The “Quote Rule” requires specialists and market makers to provide quotation information, or a quote, that must be the best prices at which he is willing to trade. This quote is therefore the lowest price the dealer will accept from a customer to sell the securities and the highest price the dealer will pay a customer to purchase the securities. A nice discussion of market infrastructure details with definitions of various terms, can be found in Hasbrouck, Sofianos and Sosebee (1993) as well as Rule 11Ac1-1 regarding the Dissemination of Quotations at <http://taft.law.uc.edu/CCL/34ActRIs/rule11Ac1-1.html> (<http://taft.law.uc.edu/CCL/34ActRIs/rule11Ac1-1.html>).

#### NBBO Algorithm:

An NBBO is the National Best Bid and Offer information which is the highest bid and lowest offer of all prevailing quotes issued by various market makers in national exchanges. The Securities and Exchange Commission uses the NBBO information to require brokers to execute customer trades at those best available prices (See Regulation NMS, 2005). The macro below allows researchers to determine the NBBO using TAQ consolidated quote data. TAQ quote data provides the consolidated tape of quotes from different market makers and is sorted by time for each symbol, to the second. Even within the same second, quotes from various market makers are provided in chronological order, and each new quote supersedes the previous quote from the same exchange.

All equities exchanges have automated display systems which are designed to display in the quote the best bids and offers in their limit order book (see Rule11Ac1-4 regarding Limit Orders: <http://taft.law.uc.edu/CCL/34ActRIs/rule11Ac1-4.html> (<http://taft.law.uc.edu/CCL/34ActRIs/rule11Ac1-4.html>)). Every new quote made by a market maker is a revision of the prevailing quote and supersedes the previous quote, either because the previous quote was consummated by a trade, or if some of the existing bids and asks were modified or expired by the market participants themselves, or if there was an influx of new bids and new offers from market participants within that exchange.

The algorithm calculates the inside quote by determining whether a given quote is eligible for NBBO consideration, before it is added to the list of prevailing quotes by exchange. This algorithm calculates the last NBBO every second; given that WRDS does not currently receive the quotes feed with millisecond timestamps. The quotes are screened by following closely the quote condition (see definition of the TAQ MODE variable) and the instructions in TickData.com paper to determine the quotes that are eligible for inclusion in the NBBO calculation in each exchange. “MODE” filters are applied within the data step, rather than in the “where” statement to ensure that prevailing eligible quotes incorporate trading halts, cancelled quotes and other

conditions described in the MODE definition. Also, quotes containing negative bid or ask prices and quotes with zero depth are deleted. The algorithm also removes crossed and locked quotes before calculating NBBO.

After elimination of non-qualified quotes, the program identifies the inside quote at every second and aggregates the depths from all the markets and/or market-makers that are quoting at that level. Using the list of all prevailing quotes by exchange, the NBBO is derived as the best (maximum) bid price, and the best (minimum) offer price. The last step of the NBBO calculation is to take the maximum prevailing bid (offer) sizes for quotes at the best bid (offer) price, using the list of prevailing quote by exchange. The inside quote is saved in the variables BB (best bid), BO (best offer), BBSize (aggregate depth at the best bid), and BOSize (aggregate depth at the best offer). The code also produces the number of exchanges with prevailing quotes that are valid for inclusion in NBBO calculation. The SAS macro below provides all the pertinent details.



```

/*
*****
*/
/* ***** WRDS RESEARCH MACROS ***** */
/*
*****
*/
/* WRDS Macro:
NBBO
/* Summary : Simplistic NBBO Derivation using SAS
Views
/* Date : September 7,
2010
/* Author : RabiH Moussawi,
WRDS
/* Variables : - YYYYMMDD is the Date Stamp for the Quote
dataset
/* - OUTSET is the output
dataset
/*
*****
*/

%MACRO NBBO (YYYYMMDD=19930104,OUTSET=nbbo);
options nonotes;
%put ;
%put ### START NBBO Calculation for: &YYYYMMDD ;

/* Preliminary: Create an Informat to Convert EX to Numeric */
/* Keep only Known Exchange Types: as Defined in TAQ Manual */
proc format;
  invalue ex_to_exn /* informat to be used in an INPUT function */
    'A'=01 /*AMEX*/
    'N'=02 /*NYSE*/
    'B'=03 /*BOST*/
    'P'=04 /*ARCA*/
    'C'=05 /*NSX -National (Cincinnati) Stock Ex*/
    'T'=06 /*NASDAQ*/
    'Q'=07 /*NASDAQ*/
    'D'=08 /*NASDAQ-ADF*/
    'X'=09 /*PHIL-NASDAQ OMX PSX*/
    'I'=10 /*ISE */
    'M'=11 /*CHIC*/
    'W'=12 /*CBOE*/
    'Z'=13 /*BATS*/
    'Y'=14 /*BATS Y-Ex*/
    'J'=15 /*DEAX-DirectEdge A*/
    'K'=16 /*DEXX-DirectEdge X*/
    otherwise=17 /* you can drop those quotes in the if statement */
  ;
run;
/* Additional Exchange Values */
/* 'E'=17 /*SIP -Market Independent (SIP - Generated)*/
/* 'S'=18 /*Consolidated Tape System*/

/* First, Generate Last Prevailing Quote for Each Exchange, every second */
data _quotes / view=_quotes;
  set taq.cq &yyyymmdd;
  by symbol date time NOTSORTED ex; length EXN 3.;
/* Convert EX to EXN for easy array reference */
EXN=input(ex,ex_to_exn.);

```

```

/* Keep the last prevailing Quote by exchange from consecutive quotes every
second */
if last.EX and 1 <= EXN <= 17;
label EXN="Exchange Code (numeric)";
drop EX MMID;
run;

/* Second, Derive NBBO from Prevailing Quotes Across Exchange */
data &outset (sortedby= SYMBOL DATE TIME index=(SYMBOL)
              label="WRDS-TAQ NBBO Data");
set _quotes;
by symbol date time;
/* Retain Observations within Each Time Block */
retain nexb1-nexb17 nexo1-nexo17 sexb1-sexb17 sexo1-sexo17;
array nexb nexb:; array nexo nexo:; array sexb sexb:; array sexo sexo:;
/* Step1. Reset NBBO for each new stock and at open and close */
if first.date or (lag(time)<"09:30:00"t <= time) or (lag(time) <= "16:00:00"t <
time) then
do i=1 to 17;
  nexb(i)=.; nexo(i)=.; sexb(i)=.; sexo(i)=.;
end;
/* Step2. Quote Rule: Prevailing Quote Supersedes Previous Quote */
nexb(exn)=bid;nexo(exn)=ofr;sexb(exn)=bidsiz;sexo(exn)=ofrsiz;
/* Step3. Determined if Prevailing Quotes is Eligible for NBBO */
/* See TAQ Manual pp 26 and pp 27 for MODE and NBBO Definitions */
/* Regulatory Trading Halts: MODE in (4,7,9,11,13,14,16,20,21,27,28) */
/* See TickData.com for more information on filters */
if mode not in (1,2,6,10,12,23) or bid >= ofr
then do; nexb(exn)=.; nexo(exn)=.; sexb(exn)=.; sexo(exn)=.; end;
if bid <= 0.01 or bidsiz <= 0 then nexb(exn)=.;
if ofr <= 0 or ofrsiz <= 0 then nexo(exn)=.;
/* Step4. Calculate NBBO */
BB=max(of nexb:);
BO=min(of nexo:);
/* Step5. Calculate Bid and Ofc Sizes at NBBO */
BBSIZE=0; BOSIZE=0;
do i=1 to 17;
  if BB=nexb(i) then BBSIZE=max(BBSIZE,sexb(i));
  if BO=nexo(i) then BOSIZE=max(BOSIZE,sexo(i));
end;
if missing(BB) then BBSIZE=.;
if missing(BO) then BOSIZE=.;
/* Report # of Exchanges with Qualifying Quotes */
length NUMEX 3.;
NUMEX=max(N(of nexb:),N(of nexo:));
/* Keep NBBO Information at the End of Each Second Interval */
if last.time then output;
label BB = 'Best Bid';
label BO = 'Best Offer';
label BBSIZE='Best Bid Size';
label BOSIZE='Best Offer Size';
label NUMEX='# of Exchanges with Prevailing Quotes used in the NBBO';
drop EXN MODE BID OFR nexb: nexo: i sexb: sexo: ofrsiz bidsiz;
run;

/* House Cleaning */
proc sql; drop view _quotes; quit;
options notes;
%put ### DONE. NBBO Data Saved as : &outset ; %put ;
%mend;

```

```

/*
*****
*/
/* ***** Material Copyright Wharton Research Data Services
***** */
/* ***** All Rights Reserved
***** */
/*
*****
*/

```

The macro can be invoked using as statement that contains the trading day information, and output dataset name, such as:

```
%nbbo(yyyymmdd=19970303,outset=nbbo_19970303);
```

See below for an example of how the NBBO looks like for IBM during the first few minutes of that trading day on March 3, 1997. The output dataset contains the symbol, date, time, as well as the various variables pertaining to the inside quote. The macro can be easily incorporated in a loop that appends the results of multiple days into a single consolidated output dataset. You can also use the loop described in the "TAQ SAS Programming Issues (/wrds/research/applications/microstructure/using TAQ/TAQ SAS Programming Issues.pdf)" document to properly iterate from one daily file to another and append the resulted output from the operations on each daily unit of data. The best part of this is that SAS View allows all those operations and manipulations of the data to occur in one step, and with only one read of the relevant data.

	Stock Symbol	Quote date	Quote time	Best Bid	Best Offer	Best Bid Size	Best Offer Size	# of Exchanges with Prevailing Quotes used in the NBBO
1	IBM	19970303	9:47:13	144.625	144.875	30	500	7
2	IBM	19970303	9:47:16	144.625	144.875	30	500	7
3	IBM	19970303	9:47:34	144.75	144.875	50	500	7
4	IBM	19970303	9:47:35	144.75	144.875	50	500	7
5	IBM	19970303	9:47:36	144.75	144.875	50	500	7
6	IBM	19970303	9:47:37	144.75	144.875	50	500	7
7	IBM	19970303	9:47:38	144.75	144.875	50	500	7
8	IBM	19970303	9:47:39	144.75	144.875	100	500	7
9	IBM	19970303	9:47:40	144.75	144.875	100	500	7
10	IBM	19970303	9:47:45	144.75	144.875	100	500	7
11	IBM	19970303	9:48:07	144.75	144.875	150	500	7
12	IBM	19970303	9:48:20	144.75	144.875	150	600	7
13	IBM	19970303	9:48:27	144.75	144.875	150	600	7
14	IBM	19970303	9:48:56	144.75	144.875	15	600	7
15	IBM	19970303	9:48:58	144.75	144.875	15	600	7
16	IBM	19970303	9:49:00	144.75	144.875	15	600	7
17	IBM	19970303	9:49:01	144.75	144.875	10	600	7
18	IBM	19970303	9:49:05	144.625	144.875	50	600	7
19	IBM	19970303	9:49:06	144.75	144.875	50	600	7

## References

Hasbrouck, Joel, 2010, "The Best Bid and Offer: A Short Note on Programs and Practices," Discussion Note (/wrds/research/applications/microstructure/NBBO derivation/note/index.cfm).

Hasbrouck, Joel, George Sofianos, and Deborah Sosebee, 1993, "New York Stock Exchange Systems and Trading Procedures," NYSE Working Paper # 93-01.

Regulation National Market System (NMS) Final Rule, Securities and Exchange Commission, 2005,

<http://www.sec.gov/rules/final/34-51808fr.pdf> (<http://www.sec.gov/rules/final/34-51808fr.pdf>).

TAQ 3 User's Guide, Version 1.1.9, October 2008, New York Stock Exchange, Inc.

TickData.com Technical Paper, 2009, "Deriving the NBBO from TAQ Level 1 Quotes, A Guide to Calculating the NBBO for US Equities,"

<http://www.tickdata.com/support/equity-data-support/equity-data-faq/>  
(<http://www.tickdata.com/support/equity-data-support/equity-data-faq/>).

WRDS, 2010, "A Note on the Use of PROC SQL with TAQ data on WRDS (/wrds/research/applications/microstructure/note\_proc\_sql\_taq/index.cfm)," Wharton Research Data Services.

WRDS, 2009, "Using TAQ Efficiently (/wrds/research/applications/microstructure/using\_taq/)," Wharton Research Data Services.



(<http://www.wharton.upenn.edu>)

About WRDS (<https://wrds-web.wharton.upenn.edu/wrds/about/index.cfm>)

WRDS FAQs ([https://wrds-web.wharton.upenn.edu/wrds/about/WRDS\\_FAQs.cfm](https://wrds-web.wharton.upenn.edu/wrds/about/WRDS_FAQs.cfm))

WRDS News (<https://wrds-web.wharton.upenn.edu/wrds/news/index.cfm>)

3 Ways to use WRDS ([https://wrds-](https://wrds-web.wharton.upenn.edu/wrds/about/three_ways_to_use_WRDS.cfm)

[web.wharton.upenn.edu/wrds/about/three\\_ways\\_to\\_use\\_WRDS.cfm](https://wrds-web.wharton.upenn.edu/wrds/about/three_ways_to_use_WRDS.cfm))

Account Types on WRDS ([https://wrds-web.wharton.upenn.edu/wrds/about/Account Types.cfm](https://wrds-web.wharton.upenn.edu/wrds/about/Account_Types.cfm))

Terms of Use (<https://wrds-web.wharton.upenn.edu/wrds/about/terms.cfm>)

Account Preferences ([\[web.wharton.upenn.edu/wrds/mywrds/preferences.cfm\]\(https://wrds-web.wharton.upenn.edu/wrds/mywrds/preferences.cfm\)\)](https://wrds-</a></p></div><div data-bbox=)

Info / Support Request ([\[web.wharton.upenn.edu/wrds/about/external\\\_support\\\_request.cfm\]\(https://wrds-web.wharton.upenn.edu/wrds/about/external\_support\_request.cfm\)\)](https://wrds-</a></p></div><div data-bbox=)

Privacy Policy (<https://wrds-web.wharton.upenn.edu/wrds/about/privacy.cfm>)

WRDS Demo (<https://wrds-web.wharton.upenn.edu/wrds/demo/>)

Conference Calendar (<https://wrds-web.wharton.upenn.edu/wrds/about/ConferenceCalendar.cfm>)

Best Paper Awards (<http://www.whartonwrds.com/best-paper-award-winners/>)

## Wharton Research Data Services

*Unless otherwise noted, all material is © 1993 - 2017, The Wharton School, University of Pennsylvania. All rights reserved.*

---